## INTEGRATED CIRCUITS



Greg Goodhue

June 1993



**AN440** 

#### Author: Greg Goodhue

The following program allows an 80C51 family microcontroller to load most of its code into a RAM over a serial link after power up and execute out of the RAM for normal operation. This can allow a final product to have firmware updates done by a simple diskette mailing. Such a program is often called a "bootstrap loader".


For this example, it is assumed that the code download is done via a serial communication link, although the program could be adapted to other forms of download. The comments at the beginning of the listing are intended to document the program and its use completely.

An additional comment would be that any static routines (low level

routines that are unlikely to change over time) can probably be put into the permanent program memory (on-chip or off-chip ROM or EPROM) along with the bootstrap loader to save program RAM space for other things.

The source code file for this program is available for downloading from the Philips computer bulletin board system. This system is open to all callers, operates 24 hours a day, and can be accessed with modems at 2400, 1200, and 300 baud. The telephone numbers for the BBS are: (800) 451-6644 (in the U.S. only) or (408) 991-2406.

#### BASIC DIAGRAM OF RAM CONNECTIONS FOR THE BOOT LOADER



return. Example: "/8A31<CR>"

;

;

:

; ;

;

;

;

;

;

;

;

;

;

; ;

;

#### RAM loader program for 80C51 family applications

#### Bootstrap Loader for Hexadecimal Files written by G. Goodhue, Philips Electronics ; This program allows downloading a hexadecimal program file over an ; asynchronous serial link to a code RAM in an 80C51 system. The downloaded ; code may then be executed as the main program for the system. This technique ; may be used in a system that normally connects to a host PC so that the code ; may come from a disk and thus be easily updated. The system RAM must be ; wired to the 80C51 system so that it appears as both data and program memory ; (wire the RAM normally, but use the logical AND of RD and PSEN for the ; output enable.) ; To use the bootstrap program, an Intel Hex file is sent through the serial ; port in 8-N-1 format at 9600 baud. The baud rate and format may be altered ; by making small changes in the serial port setup routine (SerStart). ; Note that there is no hardware handshaking (e.g. RTS/CTS or XON/XOFF) ; implemented between the host and the bootstrap system. This was done to keep ; the protocol between the two systems as simple as possible. ; Since the bootstrap program does not echo the data file, there is no chance ; of an overrun unless the 80C51 is running very slowly and/or the ; communication is very fast. An 80C51 running at 11.0592 MHz (the most ; commonly used frequency in systems with serial communication) will be able ; to easily keep up with 38.4K baud communication without handshaking. ; The download protocol for this program is as follows: - When the bootstrap program starts up, it sends a prompt character ("=") up the serial link to the host. - The host may then send the hexadecimal program file down the serial link. At any time, the host may send an escape character (1B hex) to abort and restart the download process from scratch, beginning from the "=" prompt. This procedure may be used to restart if a download error occurs. - At the end of a hex file download, a colon (":") prompt is returned. If an error or other suspicious circumstance occurred, a flag value will also be returned as shown below. The flag is a bit map of possible conditions and so may represent more than one problem. If an error occurs, the bootstrap program will refuse to execute the downloaded program. Exception codes: 01 - non-hexadecimal characters found embedded in a data line. 02 - bad record type found. 04 - incorrect line checksum found. 08 - no data found. 10 - incremented address overflowed back to zero. 20 - RAM data write did not verify correctly. - If a download error occurs, the download may be retried by first sending an escape character. Until the escape is received, the bootstrap program will refuse to accept any data and will echo a question mark ("?") for any character sent. - After a valid file download, the bootstrap program will send a message containing the file checksum. This is the arithmetic sum of all of the DATA bytes (not addresses, record types, etc.) in the file, truncated to 16 bits. This checksum appears in parentheses: "(abcd)". Program execution may then be started by telling the bootstrap program the

#### AN440

correct starting address. The format for this is to send a slash ("/")followed by the address in ASCII hexadecimal, followed by a carriage

## AN440

Application note

; - If the address is accepted, an at sign ("@") is returned before executing ; the jump to the downloaded file.										
; The bootstrap loader can be configured to re-map interrupt vectors to the ; downloaded program if jumps to the correct addresses are set up. For ; instance, if the program RAM in the system where this program is to be used ; starts at 8000 hexadecimal, the re-mapped interrupts may begin at 8003 for ; external interrupt 0, etc.										
;======================================										
<pre>\$Title(Bootstrap Loader for Hexadecimal Files) \$Date(04-13-92) \$MOD51</pre>										
;========			===							
;		Def	ini	itions						
;=========										
LF	EQU	0Ah	;	Line Feed character.						
CR	EQU	0Dh	;	Carriage Return character.						
ESC	EQU	1Bh	;	Escape character.						
StartChar	EQU	':'	;	Line start character for hex file.						
Slash	EQU	'/'	;	Go command character.						
Skip	EQU	13	;	Value for "Skip" state.						
Ch	DATA	0Fh	;	Last character received.						
State	DATA	10h	;	Identifies the state in process.						
DataByte	DATA	11h	;	Last data byte received.						
ByteCount	DATA	12h	;	Data byte count from current line.						
HighAddr	DATA	13h	;	High and low address bytes from the						
LowAddr	DATA	14h	;	current data line.						
RecType	DATA	15h	;	Line record type for this line.						
ChkSum	DATA	16h	;	Calculated checksum received.						
HASave	DATA	17h	;	Saves the high and low address bytes						
LASave	DATA	18h	;							
FilChkHi	DATA	19h	;	File checksum high byte.						
FilChkLo	DATA	1Ah	;	File checksum low byte.						
				_						
Flags	DATA	20h	;	State condition flags.						
HexFlag	BIT	Flags.0	;	Hex character found.						
EndFlag	BIT	Flags.1	;	End record found.						
DoneFlag	BIT	Flags.2	;	Processing done (end record or some						
			;	kind of error.						
EFlags	DATA	21h		Exception flags.						
ErrFlag1	BIT	EFlags.0	;	Non-hex character embedded in data.						
ErrFlag2	BIT	EFlags.1	;	Bad record type.						
ErrFlag3	BIT	EFlags.2	;	Bad line checksum.						
ErrFlag4	BIT	EFlags.3	;	No data found.						
ErrFlag5	BIT	EFlags.4	;	Incremented address overflow.						
ErrFlag6	BIT	EFlags.5	;	Data storage verify error.						
DatSkipFlag	BIT	Flags.3	;	Any data found should be ignored.						

Reset and Interrupt Vectors ; The following are dummy labels for re-mapped interrupt vectors. The ; addresses should be changed to match the memory map of the target system. EQU 8003h ExInt0 ; Remap address for ext interrupt 0. TOInt EQU 800Bh ; Timer 0 interrupt. 8013h ; External interrupt 1. ExTnt1 EOU TlInt EQU 801Bh ; Timer 1 interrupt. SerInt EOU 8023h ; Serial port interrupt. 0000h ORG ; Go to the downloader program. LJMP Start ; The following are intended to allow re-mapping the interrupt vectors to the ; users downloaded program. The jump addresses should be adjusted to reflect ; the memory mapping used in the actual application. ; Other (or different) interrupt vectors may need to be added if the target ; processor is not an 80C51. ORG 0003h ; External interrupt 0. LJMP ExInt0 ; RETT ORG 000Bh LJMP T0Int ; Timer 0 interrupt. RETI ORG 0013h ; LJMP ExIntl ; External interrupt 1. RETI ORG 001Bh ; LJMP TlInt ; Timer 1 interrupt. RETI ORG 0023h LJMP ; Serial port interrupt. SerInt RETI Reset and Interrupt Vectors ; ; Turn off all interrupts. Start: MOV IE,#0 ; Start stack near top of '51 RAM. MOV SP,#5Fh ACALL SerStart ; Setup and start serial port. ACALL CRLF ; Send a prompt that we are here. A,#'=' "<CRLF> =" MOM ; ACALL PutChar ACALL ; Try to read hex file from serial port. HexIn ACALL ; Send a message for any errors or ErrPrt warnings that were noted. ; We want to get stuck if a fatal MOV A.EFlags JΖ HexOK ; error occurred. A,#'?' ErrLoop: MOV ; Send a prompt to confirm that we PutChar are 'stuck'. " ? ' ACALL ; ACALL GetChar ; Wait for escape char to flag reload. SJMP ErrLoop HexOK: MOV EFlags,#0 ; Clear errors flag in case we re-try. ; Look for GO command. ACALL GetChar CJNE A, #Slash, HexOK ; Ignore other characters received. ACALL ; Get the GO high address byte. GetByte ErrFlag1,HexOK ; If non-hex char found, try again. JB MOV HighAddr, DataByte ; Save upper GO address byte. ACALL GetBvte ; Get the GO low address byte. ErrFlag1,HexOK ; If non-hex char found, try again. JB MOV LowAddr,DataByte ; Save the lower GO address byte. ACALL GetChar ; Look for CR. CINE A, #CR, HexOK ; Re-try if CR not there.

#### AN440

AN440

Application note

```
; All conditions are met, so hope the data file and the GO address are all
   correct, because now we're committed.
                                   ; Send confirmation to GO. " @ "
           MOV
                   A,#'@'
           ACALL
                   PutChar
           JNB
                   ΤΙ, Ś
                                   ; Wait for completion before GOing.
           PUSH
                   LowAddr
                                   ; Put the GO address on the stack,
           PUSH
                   HighAddr
                                   ;
                                       so we can Return to it.
                                    ; Finally, go execute the user program!
           RET
;
                       Hexadecimal File Input Routine
HexIn:
          CLR
                                   ; Clear out some variables.
                   Α
           MOV
                  State,A
          MOV
                  Flags,A
                   HighAddr,A
           MOV
          MOV
                   LowAddr,A
           MOV
                  HASave,A
           MOV
                  LASave,A
           MOV
                   ChkSum,A
           MOV
                   FilChkHi,A
                   FilChkLo,A
           MOV
           MOV
                   EFlags,A
           SETB
                                   ; Start with a 'no data' condition.
                   ErrFlaq4
StateLoop: ACALL
                                   ; Get a character for processing.
                   GetChar
           ACALL
                   AscHex
                                    ; Convert ASCII-hex character to hex.
                                   ; Save result for later.
           MOV
                   Ch,A
           ACALL
                   GoState
                                   ; Go find the next state based on
                                   ;
                                       this char.
           JNB
                   DoneFlag, StateLoop ; Repeat until done or terminated.
                                   ; Send the file checksum back as
           ACALL
                   PutChar
           MOV
                   A,#'('
                                    ; confirmation. " (abcd) "
                   PutChar
           ACALL
           MOV
                   A,FilChkHi
           ACALL
                   PrByte
           MOV
                   A,FilChkLo
           ACALL
                   PrByte
                   A,#')'
           MOV
                   PutChar
           ACALL
           ACALL
                   CRLF
           RET
                                    ; Exit to main program.
; Find and execute the state routine pointed to by "State".
GoState:
          MOV
                   A,State
                                    ; Get current state.
                                   ; Insure branch is within table range.
          ANL
                   A,#0Fh
           RL
                   А
                                    ; Adjust offset for 2 byte insts.
           MOV
                   DPTR,#StateTable
           JMP
                   @A+DPTR
                                   ; Go to appropriate state.
                                   ; 0 - Wait for start.
StateTable: AJMP
                   StWait
           AJMP
                   StLeft
                                    ;
                                      1 - First nibble of count.
                                   ; 2 - Get count.
           AJMP
                   StGetCnt
           AJMP
                   StLeft
                                   ; 3 - First nibble of address byte 1.
           AJMP
                   StGetAd1
                                   ; 4 - Get address byte 1.
                                      5 - First nibble of address byte 2.
           AJMP
                   StLeft
                                   ;
                                   ; 6 - Get address byte 2.
           AJMP
                   StGetAd2
                   StLeft
                                   ; 7 - First nibble of record type.
           AJMP
           AJMP
                   StGetRec
                                   ; 8 - Get record type.
           AJMP
                                   ; 9 - First nibble of data byte.
                   StLeft
                                   ; 10 - Get data byte.
           AJMP
                   StGetDat
                                   ; 11 - First nibble of checksum.
           AJMP
                   StLeft
           AJMP
                   StGetChk
                                   ; 12 - Get checksum.
           AJTMP
                   StSkip
                                   ; 13 - Skip data after error condition.
                                   ; 14 - Should never get here.
           AJMP
                   BadState
                                    ; 15 -
                                                  "
           AJMP
                   BadState
```

#### Application note

#### AN440

; This state is used to wait for a line start character. Any other characters received prior to the line start are simply ignored. MOV StWait: A,Ch ; Retrieve input character. A, #StartChar, SWEX ; Check for line start. CJNE INC State ; Received line start. SWEX: RET ; Process the first nibble of any hex byte. StLeft: MOV A,Ch ; Retrieve input character. HexFlag,SLERR ; Check for hex character. JNB ANL A,#0Fh ; Isolate one nibble. SWAP ; Move nibble too upper location. А MOV DataByte,A ; Save left/upper nibble. INC State ; Go to next state. ; Return to state loop. RET ; Error - non-hex character found. SLERR: SETB ErrFlaq1 SETB DoneFlag ; File considered corrupt. Tell main. RET ; Process the second nibble of any hex byte. StRight: MOV A.Ch ; Retrieve input character. HexFlag,SRERR ; Check for hex character. JNB A,#0Fh ; Isolate one nibble. ANL ORL A,DataByte ; Complete one byte. MOV DataByte,A ; Save data byte. ADD A,ChkSum ; Update line checksum, MOV ChkSum,A ; and save. RET ; Return to state loop. SETB SRERR: ErrFlag1 ; Error - non-hex character found. SETB DoneFlag ; File considered corrupt. Tell main. RET ; Get data byte count for line. StGetCnt: ACALL StRight ; Complete the data count byte. MOV A,DataByte ByteCount,A MOV INC State ; Go to next state. RET ; Return to state loop. ; Get upper address byte for line. StGetAd1: ACALL StRight ; Complete the upper address byte. MOV A,DataByte MOV HighAddr,A ; Save new high address. INC ; Go to next state. State RET ; Return to state loop. ; Get lower address byte for line. StGetAd2: ACALL StRight ; Complete the lower address byte. MOV A,DataByte MOV LowAddr,A ; Save new low address. INC State ; Go to next state. RET ; Return to state loop. ; Get record type for line. ACALL ; Complete the record type byte. StGetRec: StRight A,DataByte MOV MOV RecType,A ; Get record type. ; This is a data record. JΖ SGRDat CJNE A,#1,SGRErr ; Check for end record. SETB EndFlag ; This is an end record. ; Ignore data embedded in end record. SETB DatSkipFlag MOV State,#11 ; Go to checksum for end record. SJTMP SGREX

Application note

## AN440

SGRDat: SGREX:	INC RET	State		Go to next state. Return to state loop.				
SGRErr:	SETB SETB RET	ErrFlag2 DoneFlag		Error, bad record type. File considered corrupt. Tell main.				
; Get a data byte.								
StGetDat:		StRight DatSkipFlag,SGD1		Complete the data byte. Don't process the data if the skip flag is on.				
	ACALL	Store		Store data byte in memory.				
	MOV ADD MOV CLR ADDC	A,DataByte A,FilChkLo FilChkLo,A A A,FilChkHi	; ; ;	Update the file checksum, which is a two-byte summation of all data bytes.				
	MOV	FilChkHi,A						
SGD1:	MOV DJNZ INC SJMP	A,DataByte ByteCount,SGDEX State SGDEX2		Last data byte? Done with data, go to next state.				
SGDEX: SGDEX2:	DEC RET	State		Set up state for next data byte. Return to state loop.				
; Get check	sum.							
StGetChk:	ACALL JNB SETB SJMP	StRight EndFlag,SGCl DoneFlag SGCEX	; ;	Complete the checksum byte. Check for an end record. If this was an end record, we are done.				
SGC1:	MOV JNZ MOV MOV MOV MOV	A,ChkSum SGCErr ChkSum,#0 State,#0 LASave,LowAddr HASave,HighAddr	; ; ;	Get calculated checksum. Result should be zero. Preset checksum for next line. Line done, go back to wait state. Save address byte from this line for later check.				
SGCEX:	RET	IIASave, IIIgiiAddi		Return to state loop.				
SGCErr:	SETB SETB RET	ErrFlag3 DoneFlag		Line checksum error. File considered corrupt. Tell main.				
; This state	e used to	skip through any a	ıda	ditional data sent, ignoring it.				
StSkip:	RET		;	Return to state loop.				
; A place t	o go if a	n illegal state com	nes	s up somehow.				
BadState:	MOV RET	State,#Skip	; ;	If we get here, something very bad happened, so return to state loop.				
; Store - S	ave data i	byte in external RA	M	at specified address.				
Store:	e: MOV DPH,HighAddr MOV DPL,LowAddr MOV A,DataByte MOVX @DPTR,A	DPL,LowAddr	;	Set up external RAM address in DPTR.				
		-	;	Store the data.				
	MOVX CJNE	A,@DPTR A,DataByte,StoreEr		Read back data for integrity check. ; Is read back OK?				
	CLR INC MOV MOV CLR	ErrFlag4 DPTR HighAddr,DPH LowAddr,DPL A	;	Show that we've found some data. Advance to the next addr in sequence. Save the new address				
	CJNE CJNE SETB		;	; Check for address overflow (both bytes are 0). Set warning for address overflow.				

Application note

AN440

StoreEx: RET StoreErr: SETB ErrFlag6 ; Data storage verify error. SETB DoneFlag ; File considered corrupt. Tell main. RET ;\_\_\_\_\_\_ Subroutines ;\_\_\_\_\_\_ ; Subroutine summary: ; SerStart - Serial port setup and start. ; GetChar - Get a character from the serial port for processing. ; GetByte - Get a hex byte from the serial port for processing. ; PutChar - Output a character to the serial port. ; AscHex - See if char in ACC is ASCII-hex and if so convert to hex nibble. ; HexAsc - Convert a hexadecimal nibble to its ASCII character equivalent. - Return any error codes to our host. ; ErrPrt - output a carriage return / line feed pair to the serial port. ; CRLF ; PrByte - Send a byte out the serial port in ASCII hexadecimal format. ; SerStart - Serial port setup and start. SerStart: MOV A, PCON ; Make sure SMOD is off. ACC.7 CLR PCON,A MOV MOV TH1,#0FDh ; Set up timer 1. TL0,#0FDh MOV TMOD,#20h MOV MOV TCON,#40h MOV SCON,#52h ; Set up serial port. RET ; GetByte - Get a hex byte from the serial port for processing. ; Get first character of byte. GetByte: ACALL GetChar ; Convert to hex. ACALL AscHex MOV Ch,A ; Save result for later. ACALL StLeft ; Process as top nibble of a hex byte. ACALL GetChar ; Get second character of byte. ; Convert to hex. ACALL AscHex ; Save result for later. MOV Ch,A ACALL ; Process as bottom nibble of hex byte. StRight RET ; GetChar - Get a character from the serial port for processing. GetChar: JNB RI,\$ ; Wait for receiver flag. CLR RI ; Clear receiver flag. A,SBUF ; Read character. MOV CJNE A, #ESC, GCEX ; Re-start immediately if Escape char. LJMP Start GCEX: RET ; PutChar - Output a character to the serial port. PutChar: JNB ΤI,\$ ; Wait for transmitter flag. CLR ΤI ; Clear transmitter flag. MOV SBUF,A ; Send character. RET ; AscHex - See if char in ACC is ASCII-hex and if so convert to a hex nibble. Returns nibble in A, HexFlag tells if char was really hex. The ACC is not altered if the character is not ASCII hex. Upper and lower case letters ;

; are recognized.

## AN440

AscHex:	CJNE	A,#'0',AH1	;	Test for ASCII numbers.	
AH1:	JC	AHBad	;	Is character is less than a '0'?	
	CJNE	A,#'9'+1,AH2	;	Test value range.	
AH2:	JC	AHVal09	;	Is character is between '0' and '9'?	
0	CJNE	A,#'A',AH3		Test for upper case hex letters.	
АН3:	JC	AHBad		Is character is less than an 'A'?	
2.114.	CJNE	A,#'F'+1,AH4		Test value range.	
AH4:	JC	AHValAF	'	Is character is between 'A' and 'F'?	
	CJNE	A,#'a',AH5	;	Test for lower case hex letters.	
AH5:	JC	AHBad		Is character is less than an 'a'?	
	CJNE	A,#'f'+1,AH6		Test value range.	
АН6:	JNC	AHBad	;	Is character is between 'a' and 'f'?	
	CLR	C			
	SUBB	A,#27h	;	Pre-adjust character to get a value.	
	SJMP	AHVal09	;	Now treat as a number.	
AHBad:	CLR	HexFlag		Flag char as non-hex, don't alter.	
	SJMP	AHEX	;	Exit	
AHValAF:	CLR	C		Due addingt changeton to not a selice	
AHVal09:	SUBB CLR	A,#7 C	'	Pre-adjust character to get a value.	
Anvaluy.	SUBB	A,#'0'		Adjust character to get a value.	
	SETB	HexFlag		Flag character as 'good' hex.	
AHEX:	RET	liexi lag	'	Fiag character as good hex.	
	ICD I				
; HexAsc -	Convert a	hexadecimal nibble	e t	to its ASCII character equivalent.	
	3 3 7 7	a #0551-			
HexAsc:	ANL	A,#0Fh		Make sure we're working with only	
	CJNE	አ #0,አት µ,አ1		one nibble. Test value range.	
HA1:	JC	A,#0Ah,HA1 HAVal09		Value is 0 to 9.	
HAI.	ADD	A,#7		Value is 0 to 9. Value is A to F, extra adjustment.	
HAVal09:	ADD	A,#'0'		Adjust value to ASCII hex.	
mittaros	RET	11/11 0	,		
; ErrPrt -	Return an	error code to our	h	ost.	
<b></b>	NOT	7 11/11/			
ErrPrt:	MOV	A,#':'		First, send a prompt that we are still here.	
	CALL MOV	PutChar		still here. Next, print the error flag value if	
	JZ	A,EFlags ErrPrtEx		it is not 0.	
	CALL	PrByte	'	it is not 0.	
ErrPrtEx:	RET	TIDycc			
; CRLF - ou	itput a ca	rriage return / lin	ne	feed pair to the serial port.	
657 F .					
CRLF:	MOV	A, #CR			
	CALL	PutChar			
	MOV CALL	A,#LF PutChar			
	RET	PULCHAL			
	1(15.1				
; PrByte -	Send a by	te out the serial p	po	rt in ASCII hexadecimal format.	
PrByte:	PUSH	ACC	;	Print ACC contents as ASCII hex.	
	SWAP	A			
	CALL	HexAsc	;	Print upper nibble.	
	CALL	PutChar			
	POP	ACC		Duint lawan wikkla	
	CALL	HexAsc	'	Print lower nibble.	
	CALL RET	PutChar			
	1/12 1				
;======================================					

Philips Semiconductors and Philips Electronics North America Corporation reserve the right to make changes, without notice, in the products, including circuits, standard cells, and/or software, described or contained herein in order to improve design and/or performance. Philips Semiconductors assumes no responsibility or liability for the use of any of these products, conveys no license or title under any patent, copyright, or mask work right to these products, and makes no representations or warranties that these products are free from patent, copyright, or mask work right infringement, unless otherwise specified. Applications that are described herein for any of these products are for illustrative purposes only. Philips Semiconductors makes no representation or warranty that such applications will be suitable for the specified use without further testing or modification.

#### LIFE SUPPORT APPLICATIONS

**Philips Semiconductors** 

811 East Arques Avenue

Sunnyvale, California 94088–3409 Telephone 800-234-7381

P.O. Box 3409

Philips Semiconductors and Philips Electronics North America Corporation Products are not designed for use in life support appliances, devices, or systems where malfunction of a Philips Semiconductors and Philips Electronics North America Corporation Product can reasonably be expected to result in a personal injury. Philips Semiconductors and Philips Electronics North America Corporation customers using or selling Philips Semiconductors and Philips Electronics North America Corporation so at their own risk and agree to fully indemnify Philips Semiconductors and Philips Electronics North America Corporation for any damages resulting from such improper use or sale.